


HT6221A 發碼的接收


文件編碼：HA0040T

簡介


HT6221A 的應用電路圖



HT6221A 是 Holtek 公司生產的多功能編碼晶片，採用 PPM (Pulse Position Modulation) 進行編碼，1.12ms 為 0，2.24ms 為 1，如下圖。



每發送一個碼，HT6221A 會先送出一個 9ms 的頭碼和 4.5ms 的間隙，然後依次送出 16 位的位址碼 (18ms~36ms)、8 位元資料碼 (9ms~18ms) 和 8 位元資料反碼，如下圖。



本文主要介紹用 HT48R30A-1 來進行解碼的程式。HT6221A 通過紅外線發射管發出信號，紅外線接收管接到 HT48R30A-1 的外部中斷輸入腳。

File name : 6221receiver.asm

作者 : Kelven

說明：程式開始時，先清除 RAM 區，然後打開主中斷、外部中斷及定時計數器中斷。通過計算中斷之間的時間間隔來解碼，當接收到有效碼後，會放置在旗標位元。在主程序中判斷旗標位元是否放置位元來判斷有效碼的接收。配置選項系統時鐘為 4000kHz。

程式範例

```
include ht48r30a-1.inc
;-----
;Filename:6221RECEIVER.asm
;Function:DECODER HT6221A
;Microprocessor:HT48R30A-1
;Crystal:4MHz
;-----
data.section 'data'

card_no0      equ    [060h]      ;
card_no1      equ    [061h]
card_no2      equ    [062h]
card_no3      equ    [063h]      ;記錄最終的資料

temp          equ    [07eh]
cint          db    ?           ;記錄收資料 bit 個數的變數
```

```

count          db ?
count_buf      db ?
int_acc        db ?           ;中斷保護變數
right_tou      dbit
y_bit          dbit
;-----
code.section at 0000 'code'
    org 00h

    jmp start
    org 04h           ;中斷入口位址
    jmp do_wait
    org 08h
    jmp do_tmr       ;有 time 中斷發生

    org 20h
start:
    set  pgc         ;中斷口設為輸入狀態
    call init
    mov  a, 81h      ;fsys/4 1M
    mov  tmrc, a
    mov  a, 56
    mov  tmr, a      ;200µs 中斷一次
    mov  a, 6
    mov  intc, a
    set  tmrc.4
;-----
sleep:
    set  intc.0
    snz  y_bit       ;判斷是否有外部中斷發生？
    jmp  sleep       ;沒有外部中斷發生則繼續等待
    call decode_1    ;解碼
    snz  right_tou   ;解碼不正確，返回重來
    jmp  start
    mov  a, cint
    sub  a, 32       ;準備收資料的 bit 個數
    snz  c           ;判斷資料是否已收完
    jmp  sleep

    clr  tmrc        ;收碼完畢
    clr  intc
    cpla card_no2
    xor  a, card_no3 ;校驗所收碼是否正確
    snz  z
    jmp  error1
    jmp  start       ;正確
error1:
    jmp  start       ;錯誤
;-----

```

```

do_wait:
    clr tmrc.4
    clr intc.0
    mov int_acc, a           ;進入中斷保護,push acc 的值暫存
    mov a, count
    mov count_buf, a
    clr count
    set y_bit
over_int:
    mov a, 56
    mov tmr, a
    mov a, int_acc         ;pop acc 的值
    set tmrc.4
    set intc.0
    reti

;-----
do_tmr:
    inc count
    reti
;~~~~~
decode_1 proc
    clr intc.0
    clr y_bit
    snz right_tou
    jmp judge_touma
    mov a, count_buf
    sub a, 4
    snz c
    jmp error              ;data 值小於 200µs×4=0.8ms 判斷錯誤
    mov a, count_buf
    sub a, 13
    sz c
    jmp error              ;data 值大於 200µs×13=2.6ms 判斷錯誤
    mov a, count_buf      ;data 值大於 200µs×8=1.6ms;c=1,data=1
    sub a, 8              ;data 值小於 200µs×8=1.6ms;c=0,data=0

    rrc card_no3
    rrc card_no2
    rrc card_no1
    rrc card_no0
    inc cint                ;記錄收到的 BIT 位元數
    ret
error:
    clr count
    clr right_tou
    ret
judge_touma:
    clr right_tou          ;頭碼判斷 13.50ms
    mov a, count_buf
    
```

```

        sub a, 70                ;200*70=14ms
        sz c
        ret                    ;頭碼大於 14ms 判斷錯誤
        mov a, count_buf
        sub a, 63                ;63*200=12.6ms
        snz c
        ret                    ;頭碼小於 12.6ms 判斷錯誤
        set right_tou
        ret
decode_1 endp
;~~~~~
;Function:      init
;Purpose:      clear ram value
;Parameter:
;Return:
;Modified:     acc, status
;~~~~~
init proc
    clr intc                    ;清除中斷
    mov a, 20h
    mov mp0, a
    mov a, 5fh
    mov temp, a
ram_clr:                    ;清除 RAM
    clr r0
    inc mp0
    sdz temp
    jmp ram_clr
    ret
init endp

```

版本記錄

版本：V1.10

修改人員：張淙宸

修改日期：2010 年 5 月 4 日

修改內容：Part No. HT6221 改 HT6221A。

應用電路圖修改。